
Experimental Section

All experimental NMR spectra were recorded on a Bruker AVANCE spectrometer 

equipped with a 5mm TCI 1H/13C/15N cryoprobe operating at 500.13 MHz for 1H at 

298K. Our starting point to record the reported MFA-COSY and MFA-HMBC 

experiments was those of the conventional gradient-selected magnitude-mode COSY 

(BRUKER pulse program: cosygpqf), HMQC (BRUKER pulse program: hmqcgpqf) 

and HMBC (BRUKER pulse program: hmbcgpqf) experiments, respectively. The 

samples used in this work were: 20 mg of sucrose dissolved in 0.6 ml of D2O, 25 mg of 

erythromycin A dissolved in CDCl3, and 20 mg of strychnine dissolved in CDCl3.

Basic conditions for the MFA-COSY/RELAY-3 experiment (Figure 1A): A pre-scan 

delay of 1s and 2 scans for each one of the 128 t1 increment was used (1024 points of 

time domain in the acquisition F2 dimension). A basic two-step phase cycling was 

applied: 1=x, -x;  rec.= x, -x. The acquisition time ' was of 93 ms, the overall 

duration of the gradient (1 ms) and its recovery delay () was 1.2 ms and the gradient 

ratio G1:G2:G3:G4 used was 50:30:40:17. The equivalent MFA-COSY/RELAY and 

MFA-COSY/RELAY-2 (Figure S2A) were acquired exactly under the same conditions. 

MFA-COSY/TOCSY experiment (Figure 2A) was also recorded under the same 

conditions, using a z-filtered DIPSI pulse trains of 60 ms and a gradient ratio G1:G2:G3 

of 10:33:20. The heteronuclear MFA-HMBC/HMBC-TOCSY (Figure 3A) and MFA-

HMBC/HMBC-COSY (Figure S4A) experiments were recorded under the same 

conditions described previously, using a delay adjusted to 62.5 ms (8 Hz 



optimization). The gradient ratio G1:G2:G3 in MFA-HMBC/HMBC-TOCSY was 

30:50:17 and in MFA-HMBC/HMBC-COSY of 30:50:40. Experimental times are 

shown in Table S1.

Basic conditions for the MFA-MBOB-COSY experiment (Figure S5): A pre-scan delay 

of 1s and 8 scans for each t1 increment was used (TD2=1K, TD1=128), with an overall 

experimental time of 2 h. The and ’ delay were adjusted to 62.5 ms (8 Hz) and 

85ms, respectively. Gradient ratio used was 50:30:40:15. Jmin and Jmax for 1J filtering 

were adjusted to 120 Hz and 170 Hz respectively. 

For all homonuclear and heteronuclear NMR experiments: Because each FID is 

allocated in different memory blocks, they were individually processed in the usual way 

using a zero-filling in both F1 and F2 dimensions, up 2048 and 1024 data matrix 

respectively, and applying a non-shifted sine-bell window function in both dimensions 

prior to Fourier transformation. 



Pattern 
Experiment

Experimental 
time MFA Experiment Experimental 

Time

Comparative 
Incremental 

(in %)
COSY 4m 54s MFA-COSY/RELAY 5m 11s 6%

MFA-COSY/RELAY-2 5m 34s 14%
MFA-COSY/RELAY-3 5m 57s 21%
MFA-COSY/TOCSY 5m 31s 13%

HMQC 4m 50s MFA-HMQC/HMQC-COSY 5m 14s 8%
MFA-HMQC/HMQC-

RELAY-3 6m 3s 25%

MFA-HMQC/HMQC-TOCSY 5m 31s 14%

HMBC 5m 11s MFA-HMBC/HMBC-COSY 5m 37s 8%
MFA-HMBC/HMBC-TOCSY 5m 53s 14%

Table S1. Comparison of experimental measurement times between different pattern 

and MFA experiments. Experimental parameters: pre-scan delay of 1 s; 2 scans per t1 

increment; 128 t1 increments of 1024 points each one; sw(1H)= 10 ppm, sw(13C)=160 

ppm; FID acquisition time of 90 ms; TOCSY mixing time of 60 ms.



Figure S1: General scheme showing the incorporation of the MFA strategy in A) 

homonuclear (the example shows the gradient-enhanced COSY as a pattern experiment) 

and B) heteronuclear (the example shows the gradient-enhanced HMBC as a pattern 

experiment) NMR experiments. Grey boxes stand for additional COSY-based or 

TOCSY-based transfers appended after the conventional FID acquisition with the aim 

of collecting multiple NMR data into the same scan.



Figure S2: A) Pulse scheme of the MFA-COSY-RELAY-2 experiment. B) Separated 

2D COSY, RELAY, and RELAY-2 NMR spectra of strychnine simultaneously 

obtained from the pulse scheme of Fig. A. See experimental section for details.



Figure S3: A) Pulse scheme of the MFA-HMQC/HMQC-COSY experiment. B) 

Separated 2D HMQC, 2D HMQC-COSY and 2D HMQC-RELAY NMR spectra of 

strychnine simultaneously obtained from the pulse scheme of Fig. S3A.



Figure S4: A) Pulse scheme of the MFA-HMBC/HMBC-COSY experiment. B) 

Separated 2D HMBC and HMBC-COSY NMR spectra of strychnine simultaneously 

obtained from the pulse scheme of Fig. S4A. 



Figure S5: Pulse sequence of the MA-MBOB-COSY experiment. i is the overall 

duration of low-pass J filter. The durations into the second-order low-pass J filter 

element were F = (0, 1, 2, 1+2), 1 = (2*(1Jmin + 0.146(1Jmax – 1Jmin)))-1  1 = 

(2*(1Jmin - 0.146(1Jmax – 1Jmin)))-1.The interpulse delay in the MBOB experiment was 

optimized at 8 Hz (=1/(2*nJCH)) and ’stands for the acquisition time. The duration of 

each gradient is specified as  and the gradient ratio was set to G1:G2:G3:G4 is 

50:30:40:15. Phase cycling is 1=x, -x, x, -x;  2=x, x, -x, -x. rec.= x, -x, -x, x. 



Figure S6: A) Separated 2D HMQC, 2D HMQC-COSY, 2D HMBC and 2D HMBC-

COSY spectra of strychnine obtained simultaneously by the herein proposed MA-

MBOB-COSY pulse sequence of Fig. S5. B) Expanded areas showing one-bond 

(HMQC), two-bond (HMQC-COSY), two- and three-bond (HMBC), and two, three- 

and four-bond correlations (HMBC-COSY). 



Pulse Programs (for Topspin v1.3)

; mfa-cosyrelay
;topspin v1.3
#include <Avance.incl>
#include <Grad.incl>
"d0=3u"
"d13=4u"
1 ze
2 d1
3 50u
  p1 ph1
  d0
  d13 UNBLKGRAD
  p16:gp1
  d16
  p1 ph2
  d13
  p16:gp1
  d16
  goscnp ph31
  d13 wr #1 
  p16:gp2
  d16
  p1 ph11
  d13
  p16:gp2*-1
  d16 
  goscnp ph31
  d13 wr #2 
  p16:gp3*-1
  d16
  p1 ph11
  d13
  p16:gp3
  d16 
  goscnp ph31
  d13 wr #3 
  p16:gp4
  d16
  p1 ph11
  d13
  p16:gp4*-1
  d16 BLKGRAD
  gosc ph31
  d1 wr #4 
  lo to 3 times 2
  30u if #1
  30u if #2
  30u if #3
  30u if #4
  30u id0
  lo to 3 times td1
  



exit

ph1=0 2 
ph2=0
ph11=1  
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;p0 : f1 channel -  20 to 90 degree high power pulse
;p1 : f1 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
;NS: 1 * n
;DS: 16
;td1: number of experiments
;FnMODE: QF
;for z-only gradients:
;gpz1: 50%
;gpz2: 30%
;gpz3: 40%
;gpz4: 17%
;use gradient files:   
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100



;mf_cosy-tocsy
;topspin v1.3
#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
"p7=p6*2"
"d13=4u"
"d0=in0/2-p1*2/3.1416-4u"
"FACTOR1=(d9/(p6*115.112))/2+0.5"
"l1=FACTOR1*2"
1 ze
2 d1
3 50u
  p1 ph1
  d0
  50u UNBLKGRAD
  p16:gp1
  d16
  p1 ph2
  50u
  p16:gp1
  d16 BLKGRAD
  goscnp ph31
  50u wr #1 
  50u UNBLKGRAD
  p1 ph2
  50u
  p16:gp3
  d16
  4u pl10:f1
4 p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23 
  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25
  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23



  p6*4.111 ph25
  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23
  lo to 4 times l1
  d12 pl1:f1
  p1 ph2
  50u
  p16:gp2
  d16 BLKGRAD
  goscnp ph31
  d1 wr #2
  lo to 3 times 2
  30u if #1
  30u if #2
  30u id0
  lo to 3 times td1
exit

ph1=0 2 
ph2=0
ph23=3
ph25=1
ph31=0 2 

;pl1 : f1 channel - power level for pulse (default)
;pl10: f1 channel - power level for TOCSY-spinlock
;p1 : f1 channel -  90 degree high power pulse
;p5 : f1 channel -  60 degree low power pulse
;p6 : f1 channel -  90 degree low power pulse
;p7 : f1 channel - 180 degree low power pulse
;p17: f1 channel - trim pulse                        [2.5 msec]
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;d9 : TOCSY mixing time
;d12: delay for power switching                      [20 usec]
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
;NS: 2 * n
;DS: 8
;td1: number of experiments
;FnMODE: QF
;for z-only gradients:
;gpz1: 10%
;gpz2: 20%
;gpz3: 33%
;use gradient files:   
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100



;mf_hmbctocsy
;topspin v1.3
#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
"p5=p6*.667"
"p7=p6*2"
"p2=p1*2"
"d0=3u"
"d13=3u"
"d6=1s/(cnst13*2)"
"DELTA1=50u+p16+d16+4u"
"SCALEF=p7*2/p5+0.5"
"FACTOR1=((d9-p17*2)/(p6*64+p5))/SCALEF+0.5"
"l1=FACTOR1*SCALEF"

1 ze 
2 d1 
3 d12 pl1:f1
  p1 ph1
  d6
  p3:f2 ph4
  d0
  50u UNBLKGRAD
  p16:gp1
  d16
  p2 ph2
  50u
  p16:gp2
  d16 
  d0
  p3:f2 ph5
  goscnp ph31
  50u wr #1 
  p16:gp3
  d16
  4u pl10:f1
  (p17 ph26)

;begin MLEV17
4 (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p5 ph23)



  lo to 4 times l1 
;end MLEV17

  (p17 ph26) 
  50u
  p16:gp3*-1
  d16 BLKGRAD
  gosc  ph31 
  d1 wr #2 
  lo to 3 times 2
  10m if #1 
  10m if #2
  10m id0  
  lo to 3 times td1  
exit 
  
ph1=0 
ph11=1
ph2=0 
ph3=0 
ph4=0 2
ph5=0 0 2 2 
ph22=3 
ph23=0 
ph24=1 
ph25=2 
ph26=0 
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d6 : delay for evolution of long range couplings
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = J(XH) long range
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: QF
;for z-only gradients:
;gpz1: 30% 
;gpz2: 50%
;gpz3: 17% for C-13
;use gradient files:   
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100



;teo_mf_hmbccosy

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d0=3u"
"d13=3u"
"d6=1s/(cnst13*2)"
"DELTA1=50u+p16+d16+4u"

1 ze 
2 d1 
3 p1 ph1
  d6
  p3:f2 ph4
  d0
  50u UNBLKGRAD
  p16:gp1
  d16
  p2 ph2
  50u
  p16:gp2
  d16 
  d0
  p3:f2 ph5
  goscnp ph31
  50u wr #1 
  p16:gp3
  d16
  p1 ph11
  50u
  p16:gp3*-1
  d16 BLKGRAD
  gosc  ph31 
  d1 wr #2 
  lo to 3 times 2
  10m if #1 
  10m if #2
  10m id0  
  lo to 3 times td1  
exit 
  

ph1=0 
ph11=1
ph2=0 
ph3=0 
ph4=0 2
ph5=0 0 2 2 
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)



;pl2 : f2 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d6 : delay for evolution of long range couplings
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = J(XH) long range
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: QF



;mf-mbob-cosy

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
  
"p4=p3*2"
"p2=p1*2"
"d0=3u"
"d13=3u"
"d2=0s"
"d3=0.5s*(1/(cnst8+0.146*(cnst9-cnst8)))"    ;cnst9: jmax
"d4=0.5s*(1/(cnst9-0.146*(cnst9-cnst8)))"    ;cnst8: jmin
"d5=d3+d4"

"d6=1s/(cnst13*2)"
"d7=d2+d3+d4+d5"

"DELTA2=d7-d2-p4"
"DELTA3=d7-d3-p4"
"DELTA4=d7-d4-p4"
"DELTA5=d7-d5-p4"

1 ze 
2 d1 pl1:f1 pl2:f2
3 p1 ph1 
    
  if "cnst26 == 0"
     {
     d2
     }
  if "cnst26 == 1"
     {
     d3
     }
  if "cnst26 == 2"
     {
     d4
     }
  if "cnst26 == 3"
     {
     d5
     }
  

  p4:f2 ph1
 

  if "cnst26 == 0"
     {
     DELTA2
     }
  if "cnst26 == 1"
     {
     DELTA3



     }
  if "cnst26 == 2"
     {
     DELTA4
     }
  if "cnst26 == 3"
     {
     DELTA5
     }
 
  d6
  p3:f2 ph3
  d0
  50u UNBLKGRAD
  p16:gp1
  d16
  p2 ph2
  50u
  p16:gp2
  d16 
  d0
  p3:f2 ph4
  4u 
  p16:gp3
  d16 
  goscnp ph31
  10u wr #1 
  p16:gp4
  d16
  p1 ph11 
  10u 
  p16:gp4
  d16 BLKGRAD   
  gosc ph31
  d1 wr #2 
  lo to 3 times ns
  10u if #1
  10u if #2 
  10u id0
  lo to 2 times td1
exit 
  
ph1=0 
ph11=1
ph2=0 
ph3=0 2
ph4=0 0 2 2
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                  [3 usec]



;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d6 : delay for evolution of long range couplings
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = J(XH) long range
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: QF


